

SE2062 Project – FormLang++ DSL

Grammar Definition and Syntax Rules

form ::= "form" identifier "{" [meta_section] { form_section } [validation_section]
"}"

meta_section ::= "meta" { meta_entry }

meta_entry ::= identifier "=" value ";"

form_section ::= "section" identifier "{" { field_definition } "}"

field_definition ::= field_type identifier { attribute } ";"

field_type ::= "text"
| "textarea"
| "number"
| "email"
| "date"
| "checkbox"
| "dropdown"
| "radio"
| "password"
| "file"

attribute ::= "required"
| "pattern" "=" string
| "default" "=" value
| "min" "=" number
| "max" "=" number
| "rows" "=" number
| "cols" "=" number
| "accept" "=" string
| "options" "=" "[" value_list "]"

value_list ::= value { "," value }

validation_section ::= "validate" "{" { condition } "}"

condition ::= "if" expression "then" "error" string ";"

expression ::= identifier comparison_operator value

comparison_operator ::= "<" | ">" | "<=" | ">=" | "==" | "!="

value ::= string | number

identifier ::= letter { letter | digit | "_" }

string ::= "'" { character } "'"

number ::= digit { digit }

character ::= letter | digit | symbol | space

letter ::= "a" | ... | "z" | "A" | ... | "Z"

digit ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"

symbol ::= "-" | "_" | "." | "@" | "#" | "\$" | "%" | "&" | "*" | "/"

space ::= " "

terminals

"form", "meta", "section", "validate", "if", "then", "error",
"=", ";", "{", "}", "<", ">", "<=", ">=", "==", "!=", "[", "]", ",", "

Non terminals

form, meta_section, meta_entry, form_section, field_definition,
field_type, attribute, value_list, validation_section, condition,
expression, comparison_operator, value, identifier, string, number,
character, letter, digit, symbol, space